

Chapter 1

Disjoint Forms in Graphical User Interfaces

Sander Evers, Peter Achten, Rinus Plasmeijer¹

Abstract: Forms are parts of a graphical user interface (GUI) that show a (structured) value and allow the user to update it. Some forms express a choice between two or more (structured) values using radio buttons or check boxes. We show that explicitly modelling such a choice leads to a cleaner separation of logic and layout. This is done by extending the combinator library `FunctionalForms` with *disjoint form combinators*. To implement these, we have generalized the technique of compositional functional references which underlies the library.

1.1 INTRODUCTION

Forms are parts of a graphical user interface (GUI) that show a (structured) value and allow the user to update it. For example, the omnipresent dialogs labeled *Options*, *Settings* and *Preferences* are forms. An address book can also be considered a form. In our previous work, we have developed the combinator library `FunctionalForms`[2] for building forms in a concise, compositional way.

Many real-life forms allow a choice between two or more alternatives, some of which require extra information. For example, the form in Fig. 1.1 indicates whether the user wishes to receive a certain newsletter; if s/he does, the text entry field next to this option should contain his/her email address. If s/he does not, this text field is irrelevant (some GUIs provide a visual clue for this: the control is dimmed).

Usually, the information in such a form is processed as a product-like data structure containing the choice (e.g. as a boolean) and the extra information (e.g. as a string). However, most functional languages allow data types which are more suited to this task, namely *disjoint union types*. In Haskell[4], we would define

¹Radboud University Nijmegen, Department of Software Technology, Toernooiveld 1, 6525 ED Nijmegen, The Netherlands. {s.evers,p.achten,rinus}@cs.ru.nl




FIGURE 1.1. A ‘disjoint’ form

data *NewsLetter* = *NewsYes String* | *NewsNo*

for the type of information in the example form.

While the combinators in `FunctionalForms` previously only supported forms with product-like data structures, in this paper we extend them to enable the explicit definition of such a *disjoint form*. Rather than as a ‘yes/no’ form *and* an ‘email’ form, it can now be composed as a ‘yes, email’ form *or* a ‘no’ form. We demonstrate that this technique leads to a better separation of logic and layout in disjoint forms. For its implementation, we have generalized the *compositional functional references* which underlie the library.

This paper is organized as follows: it first gives a summary of the library’s basic use, which has not changed (Sect. 1.2). Then, the use and merits of the extension are demonstrated (Sect. 1.3), after which its implementation is discussed in Sect. 1.4. Next, we show that the gained flexibility leads to some safety issues (Sect. 1.5). We finish with related work (Sect. 1.6) and conclusions (Sect. 1.7).

1.2 FUNCTIONALFORMS SUMMARY

`FunctionalForms`[2] is a combinator library built on the GUI library `wxHaskell`[7] (itself built on the cross-platform C++ library `wxWidgets`[11]). It can be seen as an embedded domain-specific language for forms: it consists of *atomic forms* and ways to combine them into larger forms in a declarative style. In this section, we give a brief summary of its basic use, which is the same as described in [2], although the types have changed a little.

A *form* is a GUI part, residing somewhere within a dialog with *OK* and *Cancel* buttons, which is only able to display and alter a certain value. When the dialog appears, the form has an *initial value* which is provided by its environment; subsequently, the user can read and alter this value; at the end, the user closes the dialog with one of the buttons, and the form passes the *final value* back to the environment. The type of this value is called the *subject type* of the form.

Atomic forms correspond to a single control containing an editable value. Examples are a text entry field, containing a *String*, and a spin control, containing an *Int*:

$$\begin{aligned} \underline{\text{entry}}' &:: \text{Monad } m \Rightarrow \\ &[\text{Prop } (\text{TextCtrl } ())] \rightarrow \text{Ref } m \text{ String} \rightarrow \text{FForm } \text{win } m \text{ Layout} \\ \underline{\text{spinCtrl}}' &:: \text{Monad } m \Rightarrow \\ &\text{Int} \rightarrow \text{Int} \rightarrow [\text{Prop } (\text{SpinCtrl } ())] \rightarrow \text{Ref } m \text{ Int} \rightarrow \text{FForm } \text{win } m \text{ Layout} \end{aligned}$$



FIGURE 1.2. *ticketsForm*

We follow the convention that library functions are underlined, and an atomic form is named after the corresponding wxHaskell function which creates its control, but with an additional prime symbol. Every atomic form is parameterized with a list of optional properties used for customizing the control, e.g. its size and font (leaving this list empty produces reasonable defaults). Some atomic forms, like *spinCtrl'*, require additional parameters: its first two arguments indicate a minimum and maximum value.

All these parameters actually have little to do with FunctionalForms: they are directly passed to the wxHaskell control creation function. In contrast, the last parameter of both forms is specific to FunctionalForms; it is a *reference value* which relates the atomic form's subject type (*String* and *Int*, resp.) to the subject type of the top-level form. A more detailed description of the *Ref* and *FForm* types is postponed to Sect. 1.4.

To combine atomic forms into larger forms, two aspects have to be composed: layout and subject type. The former is performed by *layout combinators* like *grid'*, *margin'* and *floatLeft'*. These are based on the wxHaskell layout combinators after which they are named, but operate directly on forms.¹ For example, the two atomic forms can be put in a grid layout with some labels (see Fig. 1.2):

$$\underline{grid}' \ 5 \ 5 \ [\ [\ \underline{label}' \ \text{"name :"}, \ \underline{entry}' \ [] \ \underline{name} \] \ , \ [\ \underline{label}' \ \text{"nr. of tickets :"}, \ \underline{spinCtrl}' \ 1 \ 6 \ [] \ \underline{nr} \] \]$$

Note that the two reference values ($\underline{name} :: Ref \ m \ String$) and ($\underline{nr} :: Ref \ m \ Int$) are free variables in this expression. Also, this form's subject type is not yet established. To complete the form composition, \underline{name} and \underline{nr} are bound in a lambda expression, onto which a *subject type combinator*, namely *declare2*, is applied:

$$\begin{aligned} \underline{ticketsForm} &:: Monad \ m \Rightarrow Ref \ m \ (String, Int) \rightarrow FForm \ win \ m \ Layout \\ \underline{ticketsForm} &= \underline{declare2} \$ \lambda (\underline{name}, \underline{nr}) \rightarrow \\ &\quad \underline{grid}' \ 5 \ 5 \ [\ [\ \underline{label}' \ \text{"name :"}, \ \underline{entry}' \ [] \ \underline{name} \] \ , \ [\ \underline{label}' \ \text{"nr. of tickets :"}, \ \underline{spinCtrl}' \ 1 \ 6 \ [] \ \underline{nr} \] \] \end{aligned}$$

This 'declares' *ticketsForm*'s subject type to be $(String, Int)$, as witnessed by its actual type declaration (which can be omitted). Just like the atomic forms, *ticketsForm* can now be used as a component of a larger form. Note how this two-stage process of form construction separates the definition of layout and subject type structures, providing a great deal of freedom to the library user (see also [2]).

¹instead of on *Layout* values of widgets—for those familiar with wxHaskell




FIGURE 1.3. *contactForm*₁

Besides *declare2*, which declares a pair, the library also provides subject type combinators for tuples of higher arity and for lists.

$$\begin{aligned} \underline{\text{declare2}} &:: \text{Monad } m \Rightarrow \\ &((\text{Ref } m \ t_1, \text{Ref } m \ t_2) \rightarrow z) \quad \rightarrow \text{Ref } m \ (t_1, t_2) \rightarrow z \\ \underline{\text{declare3}} &:: \text{Monad } m \Rightarrow \\ &((\text{Ref } m \ t_1, \text{Ref } m \ t_2, \text{Ref } m \ t_3) \rightarrow z) \rightarrow \text{Ref } m \ (t_1, t_2, t_3) \rightarrow z \\ &\dots \\ \underline{\text{declareL}} &:: \text{Monad } m \Rightarrow \\ &([\text{Ref } m \ t] \rightarrow z) \quad \rightarrow \text{Ref } m \ [t] \rightarrow z \end{aligned}$$

The *declareL* combinator only composes forms for the list *elements* and cannot alter the spine; it produces a form for lists of a fixed length.

To run a form in a wxHaskell program, the library function *runInDialog* is used. For example, this runs the above defined *ticketsForm* with *John* and 2 in the atomic forms:

```
do ...
  (newname, newnr) ← runInDialog parentWindow ticketsForm ("John", 2)
  ...
```

The function takes as its arguments a pointer to a parent window, the form itself, and an initial value of the form's subject type. It returns an IO action, which produces a modal dialog containing the form and *OK/Cancel* buttons. When the user presses *OK*, the return value is bound to altered value in the form; if *Cancel* is pressed, the initial value is returned instead. After this, the IO thread continues.

1.3 COMBINATORS FOR DISJOINT FORMS

This section describes, from a library user's point of view, the additions for defining disjoint forms. As an example, we will define a form for contact information, depicted in Fig. 1.3. It has subject type

```
data Contact = ByPhone Phone | ByEmail String | NotAtAll
```

and expresses a choice between a phone number, an email address and no information at all. Before we can start defining the form itself, we need to define three custom subject type combinators for this type's data constructors. This is

done using a Template Haskell [8] macro named *declare*, which is included in the library.

```
declareByPhone = $(declare [|ByPhone|] 1)
declareByEmail = $(declare [|ByEmail|] 1)
declareNotAtAll = $(declare [|NotAtAll|] 0)
```

For each of the constructors, we provide the macro with its name and arity. The delimiters $\$(\dots)$ and $[|\dots|]$ are Template Haskell syntax, which the library user does not need to worry about.²

The three fresh subject type combinators are used to turn forms with subject types (resp.) *Phone*, *String* and no subject type at all into forms with subject type *Contact*. Their type signatures are:

```
declareByPhone :: Monad m =>
  (Ref m Phone -> z) -> Ref m Contact -> z
declareByEmail :: Monad m =>
  (Ref m String -> z) -> Ref m Contact -> z
declareNotAtAll :: Monad m =>
  FForm win m l -> Ref m Contact -> FForm win m l
```

In the last type signature, the type *FForm win m l* plays the same role as *z* in the above signatures. The reason why it is more constrained is that *declareNotAtAll* appends its argument form with an invisible form for handling the *NotAtAll* value.

Using these subject type combinators, *contactForm₁* can be defined as follows; we assume (*phoneForm* :: *Ref m Phone* -> *FForm win m Layout*) is defined somewhere else:

```
contactForm1 = radioGrid [byPhone, byEmail, byNothing]
byPhone      = declareByPhone $ \phone ->
  row' 5 [label' "by phone :", phoneForm phone]
byEmail      = declareByEmail $ \email ->
  row' 5 [label' "by email :", entry' [] email]
byNothing    = declareNotAtAll $
  label' "do not contact me"
```

The new disjoint form combinator *radioGrid* arranges its list of subforms into a grid layout, with radio buttons to the left of them. Due to their subject type combinators, the three subforms have the same subject type as the composite form (*Contact*), but each only ‘works’ for a particular data constructor. For example, the *byEmail* form only handles *ByEmail* values. This means that when *contactForm₁* is run with an initial *ByEmail* value, the middle radio button is selected, and only the text field next to it receives an initial value. The other text

²Template Haskell is a GHC compiler extension for meta-programming, i.e. programmatically manipulating a program at the syntactic level. The delimiters turn a meta-language expression into an object-language expression and vice versa. Both object language and meta-language are Haskell.

FIGURE 1.4. *contactForm₂*

FIGURE 1.5. *contactForm₃*

field is left empty (or contains a default value, if the programmer has specified this in *phoneForm*). When the form is closed, every subform contains a final value with its particular *Contact* data constructor, but only one of them is promoted to *contactForm₁*'s final value; this choice is determined by the radio button selected at that time.

What is the advantage of using the disjoint form combinator *radioGrid*, apart from stylistic arguments? Consider the alternative case, in which the form in Fig. 1.3 is defined as a *conjunction* of a *radioBox*' (with an *Int* for three possible choices), a *phoneForm* and an *entry*'; its subject type would be $(Int, Phone, String)$. At some later time, the interaction design department decides the form should rather look like Fig. 1.4 or like Fig. 1.5. Note that these forms still express exactly the same choice. However, when the form code is changed accordingly, its subject type would be $(Int, Int, Phone, String)$ or $(Bool, Int, Phone, String)$, and the code which handles the form data should also be altered.

If we use disjoint forms instead, the disjoint subject type can remain the same. In the code, we only need to add an extra *radioGrid* for the first case:

```

contactForm2 = radioGrid
  [noContact, λyes → row' 5 [label' "yes", yesContact yes]]
noContact    = declareNotAtAll $ label' "no"
yesContact   = radioGrid [byPhone, byEmail]

```

For the second case, we use another disjoint form combinator, namely *checkRow*:

```

contactForm3 = checkRow
  (λyes → column' 5 [label' "Please contact me", yesContact yes])
  (declareNotAtAll noLayout)

```

The functionality of these forms is still the same: they display a value of their subject type *Contact*, and allow the user to change it into another value of that type.

1.4 IMPLEMENTATION

Although the user of *FunctionalForms* does not notice a difference, apart from the new combinators and slightly altered *Ref* and *FForm* types, the implementation of the library has undergone substantial changes since its first version in [2]. These

allow for generalized forms, which may *fail* to consume an initial value (or produce a final value), and which can be joined with the disjoint form combinators *radioGrid* and *checkRow*. To construct these forms, the *compositional functional references* have also been generalized. Furthermore, the ‘heart’ of a form, which determines the communication with its environment, has been made explicit in a type *RefLink*. In order to deal with the new *FForm* type in Sect. 1.4.3, we will first discuss these *Ref* and *RefLink* types.

1.4.1 The *Ref* type

A reference value consists of two functions which are used to ‘refer to’ a *t* part of a—usually stateful—monad *m*:

$$\mathbf{data} \text{ Ref } m \ t = \text{Ref} \left\{ \begin{array}{l} \text{val} :: m \ t \\ \text{app} :: (t \rightarrow m \ t) \rightarrow m \ () \end{array} \right\}$$

The *val* function retrieves the value of this particular part of the monadic state, whereas the *app* function updates it. For example, for a state of type (t_1, t_2) , the value referring to the t_1 element would be:

$$\begin{aligned} \text{reffst} &:: \text{MonadState } (t_1, t_2) \ m \Rightarrow \text{Ref } m \ t_1 \\ \text{reffst} &= \text{Ref} \left\{ \begin{array}{l} \text{val} = \mathbf{do} \{ (x, y) \leftarrow \text{get}; \text{return } x \} \\ \text{app} = \lambda f \rightarrow \mathbf{do} \{ \tilde{(x, y)} \leftarrow \text{get}; x' \leftarrow f \ x; \text{put } (x', y) \} \end{array} \right\} \end{aligned}$$

Note the lazy pattern match in the *app* function; it is useful when constructing a new state from scratch (i.e. the previous state contains \perp).

A reference to the value in a *Just* data constructor (from the well-known *Maybe* type) can be defined in a very similar way:

$$\begin{aligned} \text{reffromJust} &:: \text{MonadState } (\text{Maybe } t) \ m \Rightarrow \text{Ref } m \ t \\ \text{reffromJust} &= \text{Ref} \left\{ \begin{array}{l} \text{val} = \mathbf{do} \{ \text{Just } x \leftarrow \text{get}; \text{return } x \} \\ \text{app} = \lambda f \rightarrow \mathbf{do} \{ \tilde{(\text{Just } x)} \leftarrow \text{get}; x' \leftarrow f \ x; \text{put } (\text{Just } x') \} \end{array} \right\} \end{aligned}$$

This reference value may seem ill-defined because it can ‘dangle’: when the monadic state contains *Nothing*, it does not refer to anything. However, this situation can be detected using monadic error-handling, and the control flow can be adapted. We will show how this is done in Sect. 1.4.2, when we join two *RefLinks*.

The operator \bullet composes two reference values, taking the referred part of the second value’s state as the state of the first value. For example, the following value refers to the first element of a pair within a *Just* value:

$$\text{reffst} \bullet \text{reffromJust} :: \text{MonadState } (\text{Maybe } (t_1, t_2)) \ m \Rightarrow \text{Ref } m \ t_1$$

The composition is performed by applying a monad transformer to the monad of the second reference value. This ‘adds state’ to this monad, on which the first

reference value can act. Meanwhile, properties of the original monad such as IO ability or error handling are preserved.

$$\begin{aligned}
& \bullet :: \text{Monad } m \Rightarrow \text{Ref } (\text{StateT } cx \ m) \ t \rightarrow \text{Ref } m \ cx \rightarrow \text{Ref } m \ t \\
& w \bullet v = \text{Ref} \\
& \quad \{ \text{val} = \text{val } v \gg= \text{evalStateT } (\text{val } w) \\
& \quad , \text{app} = \lambda f \rightarrow (\text{app } v) \$ \text{execStateT } \$ (\text{app } w) \$ \text{lift } .f \\
& \quad \}
\end{aligned}$$

This operator is used to define subject type combinators like:

$$\begin{aligned}
& \underline{\text{declare2}} :: \text{Monad } m \Rightarrow \\
& \quad ((\text{Ref } m \ t_1, \text{Ref } m \ t_2) \rightarrow z) \rightarrow \text{Ref } m \ (t_1, t_2) \rightarrow z \\
& \underline{\text{declare2}} \text{ refsToForm } \text{refP} = \text{refsToForm } (\text{reffst} \bullet \text{refP}, \text{refsnd} \bullet \text{refP}) \\
& \\
& \underline{\text{declareJust}} :: \text{Monad } m \Rightarrow (\text{Ref } m \ a \rightarrow z) \rightarrow \text{Ref } m \ (\text{Maybe } a) \rightarrow z \\
& \underline{\text{declareJust}} \text{ refToForm } \text{refMaybe} = \text{refToForm } (\text{reffromJust} \bullet \text{refMaybe})
\end{aligned}$$

These subject type combinators all follow the same pattern. This pattern is captured in the Template Haskell macro declare, so definitions like the two above do not have to be handwritten for every data constructor.

1.4.2 The RefLink type

The heart of an atomic form consists of a link between two reference values. The first is of type $\text{Ref } m \ t$, and relates the subject type t of this form to that of the topmost form (the state type in m). This is the reference value that is explicitly provided by the library user, e.g. in the expression $\text{entry}' \square \text{ref}_m$. The second reference value is implicit in every atomic form; it is of type $\text{Ref } IO \ t$, and relates this form's subject type to a part of the IO state. It is constructed from wxHaskell's *get* and *set* functions for the control's main attribute (e.g. *text* on an *entry* control).

In the terminology of the well-known model–view(–controller) paradigm[6], the reference values refer to a part of the topmost form's *model* and a part of its *view*, respectively. Joining them in a *RefLink* means linking those parts to each other: the *val* output from the first reference is used as *app* input for the second, and vice versa. Thus, two operations are obtained which both enforce consistency between model and view:

- The *update* operation copies the value from the form's model to its view. This is used to show the form's initial value.
- The *propagate* operation copies the value from the form's view to its model. This is used to read the form's final value.

In monadic terms, the *update* operation is a read action in the m monad producing a write action in the IO monad. Conversely, the *propagate* operation is a read

action in the *IO* monad producing a write action in the *m* monad. This results in the following type for *RefLink* (where the *n* monad abstracts from *IO*):

```
data RefLink m n = RefLink
  { update    :: m (n ())
  , propagate :: n (m ())
  }
```

The function *refLink* connects the two references to create such a *RefLink*. For the *update* function, first an input *v* is retrieved from the *m* reference. Then, a constant function *const* (*return v*) is applied to the corresponding part in the *n* monad using the *n* reference; this action in the *n* monad is returned in the *m* monad. For the *propagate* function, the roles of *m* and *n* are reversed:

```
refLink :: (Monad m, Monad n) => Ref m a -> Ref n a -> RefLink m n
refLink ref_m ref_n = RefLink
  { update    = (val ref_m) >>= return . (app ref_n) . const . return
  , propagate = (val ref_n) >>= return . (app ref_m) . const . return
  }
```

When two forms are joined, their *RefLinks* are combined into a new *RefLink*. Usually, the intention is that the joint *update* performs *both* component *updates*, and likewise for the *propagates*. We consider this to be the ‘default’ operator on *RefLink*. In order to meet the *MonadWriter* interface (see Sect. 1.4.3), we encode it using the *Monoid* class:

```
instance (Monad m, Monad n) => Monoid (RefLink m n)
  where mempty = RefLink
        { update    = return $ return ()
        , propagate = return $ return ()
        }
    rl1 `mappend` rl2 = RefLink
        { update    = liftM2 (>>) (update rl1) (update rl2)
        , propagate = liftM2 (>>) (propagate rl1) (propagate rl2)
        }
```

For disjoint forms, the two *RefLinks* should be joined in an alternative way. In this situation, they share one part of the model part, which is a disjoint union (e.g. the subject type of both forms is *Either a b*). Meanwhile, they refer to different parts of the view (which contains controls for both *a* and *b*). Hence, the two *RefLinks* connect independent parts of the view state space to ‘competing’ parts of the model state space. Instead of performing both *update* (*propagate*) operations, only one can (should) be performed.

We obtain this behaviour by using the *mplus* operator of the model monad *m*; therefore, this should be an instance of *MonadPlus*. The joint *update* will then (dynamically) choose between the first component *update* or the second—and likewise for *propagate*. Hence, we define an alternative monoid on the *RefLink*

domain. By using a different representation for the *RefLink* type, the *Monoid* class can again be used for this:

```
newtype RefLinkPlusM m n = RefLinkPlusM {pm :: RefLink m n}
instance (MonadPlus m, Monad n) => Monoid (RefLinkPlusM m n) where
  mempty = RefLinkPlusM $ RefLink
    { update    = mzero
    , propagate = return mzero
    }
  rl1 'mappend' rl2 = RefLinkPlusM $ RefLink
    { update    = (update $ pm rl1) 'mplus' (update $ pm rl2)
    , propagate = liftM2 mplus (propagate $ pm rl1) (propagate $ pm rl2)
    }
```

The exact semantics of *mplus* depend on the monad *m*. In practice, we use an error-handling state monad *ErrorT e (State a)*. This means that the first argument of *mplus* is always tried first; if it fails, the second argument is tried. When disjoint forms are used correctly, the alternatives are mutually exclusive, so this order is irrelevant.

1.4.3 The *FForm* type

A form is a value of the following type:

```
newtype FForm win m a = FForm
  { runFForm :: Window win -> IO (a, RefLink m IO) }
```

It contains three pieces of information:

1. An IO action which creates the form's controls. This action depends on a pointer to a parent window of type *Window win*, in which they are created.
2. A *RefLink* used to update the values in the controls from—and propagate them to—the form's model.
3. Additional information of type *a*; usually layout information of type *Layout* (defined by the *wxHaskell* library).

The *FForm* type can be used as a monad, which *binds* the additional (layout) information, *reads* the window pointer, *executes* the control creation functions, and *writes* a *RefLink*.

```
instance (Monad m) => Monad (FForm win m)
instance (Monad m) => MonadReader (Window win) (FForm win m)
instance (Monad m) => MonadIO (FForm win m)
instance (Monad m) => MonadWriter (RefLink m IO) (FForm win m)
```

So, $form_1 \gg= f$ means:

- f is applied to the additional information from $form_1$, producing (let's call it) $form_2$.
- The window pointer passed to $form_1 \gg= f$ is passed to $form_1$ and $form_2$.
- The IO actions in $form_1$ and $form_2$ are sequenced.
- The *RefLink* in $form_2$ is joined to the *RefLink* in $form_1$ using the 'default' *mappend* operator.

Furthermore, functions like *ask* (extract the window pointer), *liftIO* (insert an IO action at form creation time) and *tell* (insert a *RefLink*) are implemented for the *FForm* monad (being an instance of *MonadReader*, *MonadIO* and *MonadWriter*).

We stated in Sect. 1.4.2 that in order to combine two forms in a disjoint way, the *RefLinkPlusM* monoid should be used, which dynamically performs *one* of the *update/propagate* operations. Meanwhile, *both* forms should be shown: at form creation time, *both* IO actions should be performed, and *both* layout values are used. Therefore, we have implemented alternative bind and unit operators for forms: $\gg\pm$ and $return^0$. They are similar to $\gg=$ and $return$ in every respect, except that they use the *RefLinkPlusM* monoid.

$$\begin{aligned} return^0 &:: MonadPlus\ m \Rightarrow a \rightarrow FForm\ win\ m\ a \\ (\gg\pm) &:: MonadPlus\ m \Rightarrow \\ & FForm\ win\ m\ a \rightarrow (a \rightarrow FForm\ win\ m\ b) \rightarrow FForm\ win\ m\ b \end{aligned}$$

These operators are at the core of the disjoint form combinators *radioGrid* and *checkRow*, whose implementation is discussed in the next section.

1.4.4 Disjoint form combinators

A naïve disjoint form combinator would be

$$\begin{aligned} refToForm_1\ 'or'\ refToForm_2 &= \lambda ref \rightarrow \\ refToForm_1\ ref \gg\pm \lambda lay_1 &\rightarrow \\ refToForm_2\ ref \gg\pm \lambda lay_2 &\rightarrow \\ return^0\ \$\ column\ 5\ [lay_1, lay_2] \end{aligned}$$

The composite form shows both forms, while the composite *update* function performs only one of the component *update* functions—the first one that succeeds. The same goes for the composite *propagate* operation. However, the form's user has no means whatsoever to discover which *update* has been performed, or to influence which *propagate* to perform!³

The *radioGrid* combinator does provide these functions: both are fulfilled by the radio buttons. When a subform's *update* is performed, the system selects the radio button in front of it. Conversely, the form's *propagate* is only performed if

³Both *propagate* operations will succeed when performed. Due to the *mplus* semantics, the first one will always be selected.

the radio button in front of it *is* indeed selected (the user influences this during the form’s lifetime).

The nice thing is that we can express this behaviour quite elegantly in terms of *RefLink* operations. We show this by defining the somewhat simpler disjoint form combinator *alt*, which is a specialisation of *radioGrid*: it joins exactly two forms (denoted *refToForm₁* and *refToForm₂*). Assume that we can create a two-button radio group, returning a reference value *refRadio* :: *Ref IO Int* to its current selection, which can take values {0, 1}. Now we can define a *RefLink* value:

```

rl1 = refLink ref0 refRadio
  where ref0 = Ref
          { val = return 0
          , app = λf → do { 0 ← f 0; return () }
          }

```

In other words: we link *refRadio* to a reference ‘to the unchangeable number 0’, whose *val* is always 0, and whose *app* function only succeeds when the result of the function application is 0. This has the effect that the *update* operation in *rl₁* always selects the topmost radio button (and succeeds), while its *propagate* operation only succeeds when this radio button is selected.⁴

We then lift *rl₁* into a form, and join it with the first subform *form₁* using \gg , which utilizes the default (conjunctive) *mappend* operator:

```
tell rl1 >> refToForm1 ref
```

This form has the desired properties: with an *update*, the radio button is only selected if the value in *form₁* can be updated, and with a *propagate*, the value in *form₁* is only propagated if the radio button is selected. We define and use *rl₂* in a similar way, producing a second form. We finish the *alt* combinator by joining both forms with $\gg\pm$:

```

refToForm1 ‘alt’ refToForm2 = λref →
  liftIO ... >>= (laybutton1, laybutton2, refRadio) →
  let
    rl1 = ... — see above
    rl2 = ...
  in
    (tell rl1 >> refToForm1 ref) >>\pm λlayform1 →
    (tell rl2 >> refToForm2 ref) >>\pm λlayform2 →
    return0 $ grid 5 5 [[laybutton1, layform1], [laybutton2, layform2]]

```

What we have omitted in the second line goes into too much implementation detail; it is an IO action which creates the radio buttons, and returns the layout values *laybutton₁* and *laybutton₂*, as well as *refRadio*.

⁴Note that this *RefLink* does not use any model state!

The *radioGrid* combinator is a straightforward generalization of *alt* for lists. The *checkRow* combinator is also very much like *alt*, but does not show its second argument form. However, it *does* use its *RefLink*.⁵

1.5 SAFETY

The flexibility provided by compositional functional references has a downside: by omitting reference values, duplicating them, or using them in the wrong places, forms with strange behaviour can be constructed. We give some examples:

1. *declare2* $\lambda(a,b) \rightarrow \text{entry}' \ [] a$

This form never shows or changes the second element of its subject type.

2. $\lambda a \rightarrow \text{row}' \ 5 \ [\text{entry}' \ [] a, \text{entry}' \ [] a]$

This form shows its value twice, and only propagates the new value in the control on the right.

3. *declareJust* \$ *entry'* []

This form is only updated if its model contains a *Just x* value (actually, this is the desired behaviour if the form is part of a disjoint form). If it does not, all forms in the same alternative of the surrounding ‘disjoint clause’ are prevented from being updated (normally, there should be none).

4. *radioGrid* [*entry'* [], *entry'* []]

This form will always put its value in the upper entry control. However, it will propagate values from whichever entry control has its radio button selected.

To prevent the construction of these forms, the programmer can follow some rules such as:

- Every declared reference should be used exactly once.
- Every data constructor of a form’s subject type should be declared exactly once.
- References declared outside a disjoint form must not be used inside it.

Of course, it would be better if these rules would be enforced automatically, e.g. by the type system. Future research should formalize these rules.

1.6 RELATED WORK

As far as we know, the idea of explicitly using a radio button grid to combine forms in a disjoint way is new. The fact that some radio buttons make other elements (ir)relevant is recognized, but existing declarative (web) form languages

⁵Its *RefLink* is joined with a *refLink* between the value *False* and the checkbox value.

have to go out of their way to specify this. In XForms[12], it is accomplished by providing an element's `relevant` property with a Boolean expression that includes an XPath pointer to the radio button choice. In WASH/CGI[9], the programmer builds a *decision tree* (see [10]) to express which data to use when a certain radio button is selected.

A simple disjoint form combinator is already introduced in the thesis[1] from which FunctionalForms originated. However, this combinator always joins exactly two alternative forms. If the subject types of the top form and bottom form are t_1 and t_2 , resp., then the subject type of the composite form is always *Either* $t_1 t_2$. In other words, logic and layout are not separated like they are presently.

Compositional references were introduced by Kagawa[5] as a means to compose mutable data structures such as arrays. In our previous work[1, 2] we used them in a more simple form and with a different goal: to conceptually separate a form's subject type and its layout.

Closely related to compositional references are *lenses*[3], which are also pairs of accessor and modifier functions. While our approach uses a lot of 'little' references throughout the program, lenses are combined into a big lens which *is* the program; this program specifies a bidirectional transformation between model and view.

1.7 CONCLUSIONS AND FUTURE WORK

In this paper, we have identified two patterns for composing forms that edit values of a disjoint union type. The first pattern involves a list of radio buttons, and the second involves a check box. To support these patterns in the FunctionalForms library, we have introduced several new combinators.

These patterns illustrate a novel view, in which a form itself can be seen as 'disjoint'. To demonstrate the fertility of this view, we have shown that these disjoint forms exhibit a cleaner separation between logic and user interface. This makes them more flexible.

However, this flexibility comes at a price: the construction of forms with unwanted behaviour is possible. Methods for preventing this have yet to be researched.

As a further enhancement to FunctionalForms, defining forms for values of a custom Haskell type is made easier, using a Template Haskell macro. This brings the library closer to real-life use. The library version discussed in this paper will shortly be available for download.⁶

We hope to further develop the approach of programming with reference values. We believe that it can be used to construct a far wider range of interfaces in a declarative way.

⁶<http://www.cs.ru.nl/~sandr/FunctionalForms>

REFERENCES

- [1] Sander Evers. Form follows function: Editor GUIs in a functional style. Master's thesis, University of Twente, 2004. Permanently available at <http://doc.utwente.nl/fid/2101>.
- [2] Sander Evers, Peter Achten, and Jan Kuper. A functional programming technique for forms in graphical user interfaces. In *Proceedings of the 16th International Workshop on Implementation and Application of Functional Languages (IFL 2004)*, Technical Report 0408, pages 81–96. University of Kiel, 2004. Selected to appear in LNCS.
- [3] J. Nathan Foster, Michael B. Greenwald, Jonathan T. Moore, Benjamin C. Pierce, and Alan Schmitt. Combinators for bi-directional tree transformations: A linguistic approach to the view update problem. Technical Report MS-CIS-04-15, University of Pennsylvania, August 2004. An earlier version appeared in the *Workshop on Programming Language Technologies for XML (PLAN-X)*, 2004, under the title “A Language for Bi-Directional Tree Transformations”.
- [4] Simon Peyton Jones. Haskell 98 language and libraries: the revised report. *Journal of Functional Programming*, 13(1):0–255, January 2003.
- [5] Koji Kagawa. Compositional references for stateful functional programming. In *Proceedings of the second ACM SIGPLAN International Conference on Functional Programming (ICFP'97)*, volume 32(8) of *SIGPLAN Notices*, pages 217–226. ACM Press, 1997.
- [6] Glenn E. Krasner and Stephen T. Pope. A cookbook for using the model-view controller user interface paradigm in smalltalk-80. *J. Object Oriented Program.*, 1(3):26–49, 1988.
- [7] Daan Leijen. wxHaskell – a portable and concise GUI library for Haskell. In *ACM SIGPLAN Haskell Workshop (HW'04)*. ACM Press, September 2004.
- [8] Tim Sheard and Simon Peyton Jones. Template metaprogramming for Haskell. In Manuel M. T. Chakravarty, editor, *ACM SIGPLAN Haskell Workshop 02*, pages 1–16. ACM Press, October 2002.
- [9] Peter Thiemann. WASH/CGI: Server-side web scripting with sessions and typed, compositional forms. In *Proceedings of the 4th International Symposium on Practical Aspects of Declarative Languages*, volume 2257 of *LNCS*, pages 192–208. Springer-Verlag, 2002.
- [10] Peter Thiemann. An embedded domain-specific language for type-safe server-side web-scripting. Technical report, Universität Freiburg, May 2003. Available at <http://www.informatik.uni-freiburg.de/~thiemann/papers/>.
- [11] The wxWidgets home page can be found at <http://www.wxwidgets.org>.
- [12] The XForms home page can be found at <http://www.w3.org/MarkUp/Forms/>.